

# PARTIALLY ORDERED SETS IN *MACAULAY2*

DAVID COOK II, SONJA MAPES, AND GWYNETH WHEILDON

ABSTRACT. We introduce the package *Posets* for *Macaulay2*. This package provides a data structure and the necessary methods for working with partially ordered sets, also called posets. In particular, the package implements methods to enumerate many commonly studied classes of posets, perform operations on posets, and calculate various invariants associated to posets.

## INTRODUCTION.

A *partial order* is a binary relation  $\preceq$  over a set  $P$  that is antisymmetric, reflexive, and transitive. A set  $P$  together with a partial order  $\preceq$  is called a *poset*, or *partially ordered set*.

Posets are combinatorial structures that are used in modern mathematical research, particularly in algebra. We introduce the package *Posets* for *Macaulay2* via three distinct posets or related ideals which arise naturally in combinatorial algebra.

We first describe two posets that are generated from algebraic objects. The intersection semilattice associated to a hyperplane arrangement can be used to compute the number of unbounded and bounded real regions cut out by a hyperplane arrangement, as well as the dimensions of the homologies of the complex complement of a hyperplane arrangement.

Given a monomial ideal, the lcm-lattice of its minimal generators gives information on the structure of the free resolution of the original ideal. Specifically, two monomial ideals with isomorphic lcm-lattices have the “same” (up to relabeling) minimal free resolution, and the lcm-lattice can be used to compute, among other things, the multigraded Betti numbers  $\beta_{i,\mathbf{b}}(R/M) = \dim_{\mathbb{k}} \operatorname{Tor}_{i,\mathbf{b}}(R/M, \mathbb{k})$  of the monomial ideal.

In contrast to the first two examples (associating a poset to an algebraic object), we then describe an ideal that is generated from a poset. In particular, the Hibi ideal of a finite poset is a squarefree monomial ideal which has many nice *algebraic* properties that can be described in terms of *combinatorial* properties of the poset. In particular, the resolution and Betti numbers, the multiplicity, the projective dimension, and the Alexander dual are all nicely described in terms of data about the poset itself.

## INTERSECTION (SEMI)LATTICES.

A *hyperplane arrangement*  $\mathcal{A}$  is a finite collection of affine hyperplanes in some vector space  $V$ . The *dimension* of a hyperplane arrangement is defined by  $\dim(\mathcal{A}) = \dim(V)$ , and the *rank* of a hyperplane arrangement  $\operatorname{rank}(\mathcal{A})$  is the dimension of the span in  $V$  of the set of normals to the hyperplanes in  $\mathcal{A}$ .

The *intersection semilattice*  $\mathcal{L}(\mathcal{A})$  of  $\mathcal{A}$  is the set of the nonempty intersections of subsets of hyperplanes  $\bigcap_{\mathcal{H} \in \mathcal{A}'} \mathcal{H}$  for  $\mathcal{A}' \subseteq \mathcal{A}$ , ordered by reverse inclusion. We include the empty intersection corresponding to  $\mathcal{A}' = \emptyset$ , which is the minimal element in the intersection meet semilattice  $\hat{0} \in \mathcal{L}(\mathcal{A})$ . If the intersection of all hyperplanes in  $\mathcal{A}$  is nonempty,  $\bigcap_{\mathcal{H} \in \mathcal{A}} \mathcal{H} \neq \emptyset$ ,

---

2010 *Mathematics Subject Classification.* 06A06, 06A11.

*Posets* version 1.0.6 available at <http://www.nd.edu/~dcook8/files/Posets.m2>.

then the intersection meet semilattice  $\mathcal{L}(A)$  is actually a lattice. Arrangements with this property are called *central arrangements*.

Consider the non-central hyperplane arrangement  $\mathcal{A} = \{\mathcal{H}_1 = V(x + y), \mathcal{H}_2 = V(x), \mathcal{H}_3 = V(x - y), \mathcal{H}_4 = V(y + 1)\}$ , where  $\mathcal{H}_i = V(\ell_i(x, y)) \subseteq \mathbb{R}^2$  denotes the hyperplane  $H_i$  of zeros of the linear form  $\ell_i(x, y)$ ; see Figure 1(i). We can construct  $\mathcal{L}(A)$  in *Macaulay2* as follows.

```
i1 : needsPackage "Posets";
i2 : R = RR[x,y];
i3 : A = {x + y, x, x - y, y + 1};
i4 : LA = intersectionLattice(A, R);
```

Further, using the method `texPoset` we can generate  $\text{\LaTeX}$  to display the Hasse diagram of  $\mathcal{L}(A)$ , as in Figure 1(ii).




FIGURE 1. The non-central hyperplane arrangement

$$\mathcal{A} = \{\mathcal{H}_1 = V(x + y), \mathcal{H}_2 = V(x), \mathcal{H}_3 = V(x - y), \mathcal{H}_4 = V(y + 1)\}$$

and its intersection semilattice  $\mathcal{L}(A)$

A theorem of Zaslavsky [Za] provides information about the topology of the complement of hyperplane arrangements in  $\mathbb{R}^n$ . Let  $\mu$  denote the Möbius function of the intersection semilattice  $\mathcal{L}(A)$ . Then the number of regions that  $\mathcal{A}$  divides  $\mathbb{R}^n$  into is

$$r(\mathcal{A}) = \sum_{x \in \mathcal{L}(A)} |\mu(\hat{0}, x)|.$$

Moreover, the number of these regions that are bounded is

$$b(\mathcal{A}) = |\mu(\mathcal{L}(A) \cup \hat{1})|,$$

where  $\mathcal{L}(A) \cup \hat{1}$  is the intersection semilattice adjoined with a maximal element.

We verify these results for the non-central hyperplane arrangement  $\mathcal{A}$  using *Macaulay2*:

```
i5 : realRegions(A, R)
o5 = 10
i6 : boundedRegions(A, R)
o6 = 2
```

Moreover, in the case of hyperplane arrangements in  $\mathbb{C}^n$ , using a theorem of Orlik and Solomon [OS], we can recover the Betti numbers (dimensions of homologies) of the complement  $\mathcal{M}_{\mathcal{A}} = \mathbb{C}^n - \cup \mathcal{A}$  of the hyperplane arrangement using purely combinatorial data of the

intersection semilattice. In particular,  $\mathcal{M}_{\mathcal{A}}$  has torsion-free integral cohomology with Betti numbers given by

$$\beta_i(\mathcal{M}_{\mathcal{A}}) = \dim_{\mathbb{C}} \left( H_i(\mathcal{M}_{\mathcal{A}}) \right) = \sum_{\substack{x \in \mathcal{L}(\mathcal{A}) \\ \dim^{\mathbb{C}}(x) = n-i}} |\mu(\hat{0}, x)|,$$

where  $\mu(\cdot)$  again represents the Möbius function. See [Wa] for details and generalizations of this formula.

*Posets* will compute the ranks of elements in a poset, where the ranks in the intersection lattice  $\text{LA}$  are determined by the codimension of elements. Combining the outputs of our rank function with the Möbius function allows us to calculate  $\beta_0(\mathcal{M}_{\mathcal{A}}) = 1$ ,  $\beta_1(\mathcal{M}_{\mathcal{A}}) = 4$ , and  $\beta_2(\mathcal{M}_{\mathcal{A}}) = 5$ .

```
i7 : RLA = rank LA
o7 = {{ideal 0}, {ideal(x+y), ideal(x), ideal(x-y), ideal(y+1)},
      {ideal(y,x), ideal(y+1,x-1), ideal(y+1,x), ideal(y+1,x+1)}}
i8 : MF = moebiusFunction LA;
i9 : apply(RLA, r -> sum(r, x -> abs MF#(ideal 0_R, x)))
o9 = {1, 4, 5}
```

#### LCM-LATTICES.

Let  $R = K[x_1, \dots, x_t]$  be the polynomial ring in  $t$  variables over the field  $K$ , where the degree of  $x_i$  is the standard basis vector  $e_i \in \mathbb{Z}^t$ . Let  $M = (m_1, \dots, m_n)$  be a monomial ideal in  $R$ , then we define the *lcm-lattice* of  $M$ , denoted  $L_M$ , as the set of all least common multiples of subsets of the generators of  $M$  partially ordered by divisibility. It is easy to see that  $L_M$  will always be a finite atomic lattice. While lcm-lattices are nicely structured, they can be difficult to compute by hand especially for large examples or for ideals where  $L_M$  is not ranked.

Consider the ideal  $M = (a^3b^2c, a^3b^2d, a^2cd, abc^2d, b^2c^2d)$  in  $R = k[a, b, c, d]$ . Then we can construct  $L_M$  in *Macaulay2* as follows. See Figure 2 for the Hasse diagram of  $L_M$ , as generated by the `texPoset` method.

```
i10 : R = QQ[a,b,c,d];
i11 : M = ideal(a^3*b^2*c, a^3*b^2*d, a^2*c*d, a*b*c^2*d, b^2*c^2*d);
i12 : LM = lcmLattice M;
```

Lcm-lattices, which were introduced by Gasharov, Peeva, and Welker [GPW], have become an important tool used in studying free resolutions of monomial ideals. There have been a number of results that use the lcm-lattice to give constructive methods for finding free resolutions for monomial ideals, for some examples see , [Cl], [PV], and [Ve].

In particular, Gasharov, Peeva, and Welker [GPW] provided a key connection between the lcm-lattice of a monomial ideal  $M$  of  $R$  and its minimal free resolution, namely, one can compute the (multigraded) Betti numbers of  $R/M$  using the lcm-lattice. Let  $\Delta(P)$  denote the order complex of the poset  $P$ , then for  $i \geq 1$  we have

$$\beta_{i,b}(R/M) = \dim \tilde{H}_{i-2}(\Delta(\hat{0}, b); k),$$

for all  $b \in L_M$ , and so

$$\beta_i(R/M) = \sum_{b \in L_M} \dim \tilde{H}_{i-2}(\Delta(\hat{0}, b); k).$$




FIGURE 2. The lcm-lattice for  $M = (a^3b^2c, a^3b^2d, a^2cd, abc^2d, b^2c^2d)$

These computations can all be done using *Posets* together with the package *Simplicial-Complexes*, by S. Popescu, G. Smith, and M. Stillman. In particular, we can show that  $\beta_{i, a^2b^2c^2d} = 0$  for all  $i$  with the following calculation.

```
i13 : D1 = orderComplex(openInterval(LM, 1_R, a^2*b^2*c^2*d));
i14 : prune HH(D1)
o14 = -1 : 0
      0 : 0
      1 : 0
o14 : GradedModule
```

Similarly, we can show that  $\beta_{1, a^3b^2cd} = 2$ .

```
i15 : D2 = orderComplex(openInterval(L, 1_R, a^3*b^2*c*d));
i16 : prune HH(D2)
o16 = -1 : 0
      2
      0 : QQ
o16 : GradedModule
```

#### HIBI IDEALS.

Let  $P = \{p_1, \dots, p_n\}$  be a finite poset with partial order  $\preceq$ , and let  $K$  be a field. The *Hibi ideal*, introduced by Herzog and Hibi [HH], of  $P$  over  $K$  is the squarefree ideal  $H_P$  in  $R = K[x_1, \dots, x_n, y_1, \dots, y_n]$  generated by the monomials

$$u_I := \prod_{p_i \in I} x_i \prod_{p_i \notin I} y_i,$$

where  $I$  is an order ideal of  $P$ , i.e., for every  $i \in I$  and  $p \in P$ , if  $p \preceq i$ , then  $p \in I$ . *Nota bene:* The Hibi ideal is the ideal of the monomial generators of the Hibi ring, a toric ring first described by Hibi [Hi].

```
i17 : P = divisorPoset 12;
i18 : HP = hibiIdeal P;
i19 : HP_*
o19 = {x x x x x x , x x x x x y , x x x x y y , x x x x y y , x x x y y y ,
```

```

0 1 2 3 4 5   0 1 2 3 4 5   0 1 2 4 3 5   0 1 2 3 4 5   0 1 3 2 4 5
x x x y y y , x x y y y y , x x y y y y , x y y y y y , y y y y y y }
0 1 2 3 4 5   0 2 1 3 4 5   0 1 2 3 4 5   0 1 2 3 4 5   0 1 2 3 4 5

```

Herzog and Hibi [HH] proved that every power of  $H_P$  has a linear resolution, and the  $i^{\text{th}}$  Betti number  $\beta_i(R/H_P)$  is the number of intervals of the distributive lattice  $\mathcal{L}(P)$  of  $P$  isomorphic to the rank  $i$  boolean lattice. Using Exercise 3.47 in Stanley’s book [St], we can recover this by looking instead at the number of elements of  $\mathcal{L}(P)$  that cover exactly  $i$  elements.

```

i20 : betti res HP
      0  1  2  3
o20 = total: 1 10 12 3
      0: 1  .  . .
      5: . 10 12 3

i21 : LP = distributiveLattice P;
i22 : cvrs = partition(last, coveringRelations LP);
i23 : iCvrs = tally apply(keys cvrs, i -> #cvrs#i);
i24 : gk = prepend(1, apply(sort keys iCvrs, k -> iCvrs#k))
o24 : {1, 6, 3}
i25 : apply(#gk, i -> sum(i..<#gk, j -> binomial(j, i) * gk_j))
o25 : {10, 12, 3}

```

Moreover, Herzog and Hibi [HH] proved that the projective dimension of  $H_P$  is the Dilworth number of  $H_P$ , i.e., the maximum length of an antichain of  $H_P$ .

```

i26 : pdim module HP == dilworthNumber P
o26 = true

```

REFERENCES.

[Cl] T. Clark, *Poset Resolutions and Lattice-Linear Monomial Ideals*, J. Algebra **323**, (2010), no. 4, 899–919.  
 [HH] J. Herzog, T. Hibi, *Distributive lattices, bipartite graphs and Alexander duality*, J. Algebraic Combin. **22** (2005), no. 3, 289–302.  
 [Hi] T. Hibi, “Distributive lattices, affine semigroup rings and algebras with straightening laws,” in *Commutative Algebra and Combinatorics, Advanced Studies in Pure Math.*, M. Nagata and H. Matsumura, (Eds.), Vol. 11, North–Holland, Amsterdam, 1987, pp. 93–109.  
 [GPW] V. Gasharov, I. Peeva, V. Welker, *The lcm-lattice in monomial resolutions*, Math. Res. Lett. **6** (1999), no. 5–6, 521–532.  
 [M2] D. R. Grayson and M. E. Stillman, *Macaulay2, a software system for research in algebraic geometry*, available at <http://www.math.uiuc.edu/Macaulay2/>.  
 [OS] P. Orlik, L. Solomon, *Combinatorics and topology of complements of hyperplanes*, Invent. Math. **56** (1980), no. 2, 167–189.  
 [PV] I. Peeva, M. Velasco. *Frames and degenerations of monomial ideals*, Trans. Amer. Math. Soc. **363** (2011), no. 4, 2029–2046.  
 [St] R. P. Stanley, “Enumerative combinatorics. Volume 1.” Second edition. Cambridge Studies in Advanced Mathematics, **49**. Cambridge University Press, Cambridge, 2012. xiv+626 pp.  
 [St2] R. P. Stanley, *An introduction to hyperplane arrangements, Geometric Combinatorics*, IAS Park City Math. Ser. Vol 13. (2007) 389–496.  
 [Ve] M. Velasco, *Minimal free resolutions that are not supported by a CW-complex*, J. Algebra **319** (2008), no. 1, 102–114.  
 [Wa] M. Wachs, *Poset topology: tools and applications, Geometric Combinatorics*, IAS Park City Math. Ser. Vol 13. (2007) 497–615.

- [Za] T. Zaslavsky, *Facing up to arrangements: face-count formulas for partitions of space by hyperplanes*, Mem. Amer. Math. Soc. **1** (1975), issue 1, no. 154, vii+102 pp.

DEPARTMENT OF MATHEMATICS, UNIVERSITY OF NOTRE DAME, NOTRE DAME, IN 46556, USA

*E-mail address:* [dcook8@nd.edu](mailto:dcook8@nd.edu)

*E-mail address:* [smapes1@nd.edu](mailto:smapes1@nd.edu)

DEPARTMENT OF MATHEMATICS, HOOD COLLEGE, FREDERICK, MD 21701, USA

*E-mail address:* [whieldon@hood.edu](mailto:whieldon@hood.edu)